Definitions
In mathematics, additive primes are prime numbers for which the sum of their decimal digits are also primes.



Task
Write a program to determine (and show here) all additive primes less than 500.

Optionally, show the number of additive primes.



Also see

   the OEIS entry:   A046704 additive primes.
   the prime-numbers entry:   additive primes.
   the geeks for geeks entry: additive prime number.
   the prime-numbers fandom: additive primes.


F is_prime(a)
   I a == 2
      R 1B
   I a < 2 | a % 2 == 0
      R 0B
   L(i) (3 .. Int(sqrt(a))).step(2)
      I a % i == 0
         R 0B
   R 1B
 
F digit_sum(=n)
   V sum = 0
   L n > 0
      sum += n % 10
      n I/= 10
   R sum
 
V additive_primes = 0
L(i) 2..499
   I is_prime(i) & is_prime(digit_sum(i))
      additive_primes++
      print(i, end' ‘ ’)
print("\nFound "additive_primes‘ additive primes less than 500’)